

Fundamentos de Inteligencia Artificial

PEC2 - 2021

Centro Asociado de la UNED en Bizkaia

Martín Romera Sobrado
Bilbao

5 de mayo de 2021

Índice

1. Introducción	2
2. Modelado del conocimiento del dominio	2
3. Descripción del sistema	2
4. Complicaciones durante el desarrollo	8

1. Introducción

En este proyecto se presenta *El Pactometro 4M Automático* basado en la herramienta de la cadena de televisión de *La Sexta* para calcular posibles pactos entre partidos políticos tras unas elecciones. Sin embargo la herramienta original carece de una característica bastante importante, que es comentar la validez del pacto calculado.

Valorar un pacto puede parecer algo muy humano, ya que un pacto es primitivamente una relación humana. Sin embargo, un pacto político también se basa en una serie de **reglas** lógicas. Por ejemplo, es más difícil que un partido de izquierdas pacte un partido de derechas, o es difícil que pacten dos partidos que han asegurado ante los medios que quieren pactar entre ellos o que solo uno de ellos quiera pactar. *El Pactometro 4M Automático* se aprovecha de este tipo de lógica para convertirse en un *Sistema Experto para la valoración de pactos tras las elecciones del 4 de Mayo en la Comunidad de Madrid*.

2. Modelado del conocimiento del dominio

El dominio sobre el que trabajará el sistema como ya se ha mencionado antes son **los pactos de las elecciones del 4M**. Para modelar este conocimiento nos aprovechamos de la información que han proporcionado los partidos a los medios de comunicación, como la intención de pacto de cada partido, o el concepto que tiene la sociedad de los partidos, como el espectro político de cada uno de estos (izquierda o derecha) y en que posición se encuentran en él (más extremos o más centristas).

Cabe decir que se ha simplificado el método de cálculo de pactos, de forma que no tiene en cuenta abstenciones, solo apoyos. Si un conjunto de partidos suma una mayoría absoluta (69 escaños o más) el sistema considerará que el pacto se podrá evaluar y considerará al líder del partido con más escaños futuro hipotético presidente.

3. Descripción del sistema

Para comenzar a introducir pactos se deberá introducir el comando `[main].`, para cargar los módulos, seguido de `init.`, donde el usuario podrá introducir al sistema los resultados sobre los que quiere trabajar.

El sistema tras introducir los datos nos dice quien habría ganado las elecciones por número de escaños. La frase con la que dice esto se elige según el número de escaños que haya obtenido ese partido (para quitarle monotoneidad al sistema).

```
?- [main].
true.

?- init.
Introduce el número de escaños que ha obtenido Unidas Podemos
|: 10.
Introduce el número de escaños que ha obtenido Más Madrid
|: 24.
Introduce el número de escaños que ha obtenido PSOE
|: 24.
Introduce el número de escaños que ha obtenido Ciudadanos
|: 0.
Introduce el número de escaños que ha obtenido Partido Popular
|: 65.
Introduce el número de escaños que ha obtenido VOX
|: 13.
el Partido Popular con 65 escaños, se sitúa como primera fuerza parlamentaria.
Para comenzar a calcular pactos, lanza el comando "pactometro."
true.
```

Figura 1: Inicialización del sistema, con los resultados electorales del 4M

Tras esto podemos introducir el comando `pactometro.` para empezar a calcular pactos. Este comando nos pedirá primero que numero de partidos va a formar el pacto y despues introducir los partidos que formarian ese pacto ¹.

Pongamos por ejemplo de pacto más posible despues de las elecciones, una coalción de las dos fuerza de derechas con representación en el parlamento , uno entre el *Partido Popular* y *VOX*. La figura 3 muestra el resultado de la ejecución del `pactometro` para ese posible pacto.

Obviamente este no es el único pacto que forma mayoría. Digamos que todos los partidos se alian en contra del *Partido Popular*. La respuesta del *Pactometro* sería algo mal educada pero acorde a lo que le hemos pedido que nos valore, como podemos ver en la figura 4.

El *Pactometro* esta diseñado para responder con una educación proporcional al sentido que tenga el pacto que se le proponga.

¹Ojo, el `pactometro` no tiene en cuenta el concepto de abstención, las abstenciones se tienen que establecer como posiciones a favor o en contra dependiendo de si esa abstención habilitaría o deshabilitaría un pacto

```
ganador(Partido) :-  
    escanos(Partido, X),  
    PRG is X mod 3,  
    nombredet(Partido, Nombre),  
    candidato(Partido, Candidato),  
    % Selección de frase pseudoaleatoria según el número de escaños  
    (  
        (PRG == 0,  
            write('Con estos resultados, '),  
            write(Candidato),  
            write(' habría ganado las elecciones.'),  
            !  
        );  
        (PRG == 1,  
            write('Ahora mismo con los resultados que tenemos, '),  
            write(Nombre),  
            write(' tendría mayoría parlamentaria en la Asamblea de Madrid.'),  
            !  
        );  
        (PRG == 2,  
            write(Nombre),  
            write(' con '),  
            write(X),  
            write(' escaños, se sitúa como primera fuerza parlamentaria.'),  
            !  
        )  
    ).
```

Figura 2: Elección de frase aleatoria

Pongamos un ejemplo más, en este caso vamos a introducir unos datos de un sondeo que hizo *El Plural*, que daría la victoria al bloque de izquierdas con los resultados del cuadro 1.

El pacto con más posibilidades sería el de los 3 partidos del bloque de izquierdas (*Unidas Podemos*, *Más Madrid* y *PSOE*). Si le planteamos este pacto al *Pactometro* nos dará como resultado la frase de la figura 5

Como el pacto es menos posible ya que el candidato del *PSOE* dijo, que no tenía intención

```

?- pactometro.
Cuantos partidos va a formar el pacto:
|: 2.
2 restantes.
Introduce un partido de usando el siguiente código:
up : Unidas Podemos
mm : Más Madrid
psoe : PSOE
cs : Ciudadanos
pp : Partido Popular
vox : VOX
|: pp.
1 restantes.
Introduce un partido de usando el siguiente código:
up : Unidas Podemos
mm : Más Madrid
psoe : PSOE
cs : Ciudadanos
pp : Partido Popular
vox : VOX
|: vox.
Una coalición entre Partido Popular, y VOX daría paso a un gobierno de derechas
↪ en la Asamblea de Madrid liderado por Isabel Díaz Ayuso con un respaldo de
↪ 78 escaños.
true .

```

Figura 3: Resultado del pactometro entre los dos partidos de derechas

```

Tú fumas crack
true .

```

Figura 4: Resultado del pactometro para un pacto muy improbable (*Unidas Podemos, Más Madrid, PSOE y VOX*)

de pactar con *Unidas Podemos*, el *Pactometro* nos lo hace saber.

Para calcular el sentido del pacto, este utiliza los siguientes mecanismos:

- Comprobar que se encuentran en el mismo espectro.
- Comprobar si quieren pactar entre ellos.

<i>Unidas Podemos</i>	10
<i>Más Madrid</i>	26
<i>PSOE</i>	33
<i>Ciudadanos</i>	0
<i>PP</i>	57
<i>VOX</i>	10

Cuadro 1: Sondeo de El Plural

Un pacto entre Unidas Podemos, Más Madrid, y PSOE podrían dar paso a un
 ↪ gobierno progresista liderado por Ángel Gabilondo con 69 escaños de
 ↪ respaldo. Sin embargo Ángel Gabilondo ha advertido que no tenía intención
 ↪ de pactar con Pablo Iglesias lo que podría bloquear esta posibilidad.
 true .

Figura 5: Respuesta a los resultados del sondeo con un pacto del bloque de izquierdas

- Calcular la mayor distancia dentro del espectro político.

Todo este calculo lo realiza el modulo `pactos.pl`. Veamos paso a paso como desarrolla este calculo el sistema. Una vez introducido los partidos que van a formar el pacto se llama a la regla que se muestra en la figura 6.

Esta regla primero elimina los partidos que no han obtenido escaños y los que se encuentran repetidos en la lista. Tras eso comprueba si el conjunto de partidos tiene mayoría o no. Después llega a una estructura similar a la que en programación imperativa podría ser un *switch/case* en el que comprueba primero si solo hay un partido con mayoría en la coalición, si no comprueba si los partidos quieren pactar entre ellos en ambas direcciones, si no comprueba si al menos una de las partes del pacto querría participar en el pacto (como podría ocurrir en un pacto *PSOE* con *Unidas Podemos*) y finalmente si hay si hay partidos dentro del hipotético pacto que no quiere pactar entre ellos. Cada una de estas situaciones llevaría a una regla de comentario que generaría un comentario acorde al pacto. Para realizar estas evaluaciones se aprovecha de unas reglas auxiliares definidas en la figura 7.

Todos estos comentarios se encuentran dentro del modulo `comentarios.pl`. Para evaluar mayorías, numero de escaños y posición en el espectro político se utilizan las reglas y constantes del modulo `partidos.pl`. Este modulo también define algunas constantes interesantes para la generación de comentarios.

```

valorar_pacto(Ps) :-
    eliminar_repes(Ps, Parts),
    eliminar_sin_escanos(Parts, Partidos),
    (
        mayoria(Partidos),
        (
            (
                % Solo hay un partido en el pacto
                solo_uno(Partidos),!,
                comen_mayoria_absoluta(Partidos)
            );
            (
                % Todos los partidos quieren pactar entre si
                not(solo_uno(Partidos)),
                pactos_bidireccionales(Partidos),!,
                comen_pactos_bidireccionales(Partidos)
            );(
                % Algunos partidos no quieren pactar con otros que a su vez
                % quieren pactar con ellos
                not(solo_uno(Partidos)),
                not(pactos_bidireccionales(Partidos)),
                pactos_unidireccionales(Partidos),!,
                comen_pactos_unidireccionales(Partidos)
            );(
                % Hay partidos que no quieren votar entre ellos
                not(solo_uno(Partidos)),
                not(pactos_bidireccionales(Partidos)),
                not(pactos_unidireccionales(Partidos)),!,
                comen_no_pactos(Partidos)
            )
        );
        % No cumplen la condición de mayoria
        not(mayoria(Partidos)),
        comen_no_mayoria(Partidos)
    ).

```

Figura 6: Regla para la valoración del pacto


```
solo_uno([_|[]]).

pactos_bidireccionales(Partidos) :-
    forall(member(X,Partidos),pacta_resto(X,Partidos)).

pacta_resto(X,Partidos) :-
    forall(member(Y,Partidos),pactaria_con(X,Y)).

pactos_unidireccionales(Partidos) :-
    forall(member(X,Partidos),pacta_uniresto(X,Partidos)).

pacta_uniresto(X,Partidos) :-
    forall(member(Y,Partidos),pacta_alguien(X,Y)).

pacta_alguien(X,Y) :-
    pactaria_con(X,Y);
    pactaria_con(Y,X).
```

Figura 7: Reglas auxiliares

4. Complicaciones durante el desarrollo

La principal dificultad del desarrollo de este proyecto ha sido la utilización de un nuevo lenguaje de un paradigma al que no estoy acostumbrado con poco tiempo de margen (culpa mía por haber empezado tarde). Es por eso que hay algunas cosas dentro del código que igual podrían mejorarse. La optimización del código igual ha sido lo más difícil, todavía no me manejo del todo bien con la orden de corte y la he utilizado básicamente para que no entrara en bucles tontos como en los algoritmos de ordenación, en la estructura tipo `switch/case` de la valoración del pacto y poco más. Puede que se pudiera haber optimizado mejor de otra forma, pero tal vez necesite más práctica con el lenguaje para verlo con mayor claridad.

La implementación modular que he utilizado al principio del desarrollo también ha sido un dolor de cabeza. Es por eso que hay unas pocas reglas donde no deberían estar. Sin embargo al final me he hecho más o menos al sistema y he podido finalizar el proyecto de forma modular para que sea más fácil leerlo.

Como he dicho antes el tiempo ha sido otro problema que ha hecho que algunas características se hayan quedado fuera del proyecto final. Aquí listo algunas de ellas:

- Poder marcar abstenciones.
- Comentarios más precisos para la evaluación de pactos concretos.
- Mayor sensación de aleatoriedad con las respuestas.
- Evaluación de votos. (Algo que afecta a estos resultados de hecho ya que el sistema reconoce que *PSOE* y *Más Madrid* tienen el mismo apoyo cuando *Más Madrid* tiene más votos).
- Comparación de los resultados con los de las pasadas elecciones.
- Generación automática del pacto más posible.

De todas formas, a pesar de las limitaciones el sistema logra dar unos resultados con sentido.

Otra complejidad del desarrollo aunque no ligada al propio sistema si no más al tema, es que es un tema “sensible”. Son unas elecciones que acaban de ocurrir, y que han estado muy polarizadas y han tenido una participación masiva. Añadirle un toque de humor a un tema como este no ha sido fácil, ya que no quería mostrar ni mis ideas políticas ni ofender a nadie. Espero haber cumplido con esto, y si no ha sido así pido disculpas por adelantado.

Listados del código

main.pl

```

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
%
%   main.pl   -   Programa a ejecutar
%
%
% author: Martín "n3m1.sys" Romera Sobrado
%
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

:- [partidos, pactos, comentarios, util].

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% Introducir los resultados
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
init :-
    pedir_escanos(up),
    pedir_escanos(mm),
    pedir_escanos(psoe),
    pedir_escanos(cs),
    pedir_escanos(pp),
    pedir_escanos(vox),
    assert(lista_partidos([])),
    forall(con_escanos(X), anadir_partido(X)),
    lista_partidos(L),
    quicksort(L,R),
    cabeza(R,Ganador),
    ganador(Ganador),
    nl,
    write('Para comenzar a calcular pactos, lanza el comando "pactometro."').

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% Empezar a calcular escanos
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
pactometro :-
    assert(lista_pacto([])),
    write('Cuántos partidos va a formar el pacto: '), nl,
    read(N),

```

```

    pedir_partido(N),
    lista_pacto(L),
    valorar_pacto(L),
    retract(lista_pacto(_)).

pedir_partido(0) :- !.
pedir_partido(N) :-
    write(N),
    write(' restantes. '),nl,
    write('Introduce un partido de usando el siguiente código: '),nl,
    write('up : Unidas Podemos'),nl,
    write('mm : Más Madrid'),nl,
    write('psoe : PSOE'),nl,
    write('cs : Ciudadanos'),nl,
    write('pp : Partido Popular'),nl,
    write('vox : VOX'),nl,
    read(X),
    (X == up; X == mm; X == psoe; X == cs; X == pp; X == vox),
    lista_pacto(L),
    forall(lista_pacto(L),retract(lista_pacto(L))),
    append(L,[X],R),
    assert(lista_pacto(R)),
    I is N - 1,
    pedir_partido(I);
    (X \= up, X \= mm, X \= psoe, X \= cs, X \= pp ,X \= vox),
    write('(!): Entrada invalida'),
    pedir_partido(N).

```

partidos.pl

```

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
%                                                                 %
%  partidos.pl    -   Cosas relacionadas con los partidos      %
%                                                                 %
% author: Martín "n3m1.sys" Romera Sobrado                     %
%                                                                 %
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

```

```

% Definición de los partidos principales %
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
nombre(up, 'Unidas Podemos').
nombre(mm, 'Más Madrid').
nombre(psoe, 'PSOE').
nombre(cs, 'Ciudadanos').
nombre(pp, 'Partido Popular').
nombre(vox, 'VOX').

nombredet(up, 'Unidas Podemos').
nombredet(mm, 'Más Madrid').
nombredet(psoe, 'el PSOE').
nombredet(cs, 'Ciudadanos').
nombredet(pp, 'el Partido Popular').
nombredet(vox, 'VOX').

candidato(up, 'Pablo Iglesias').
candidato(mm, 'Mónica García').
candidato(psoe, 'Ángel Gabilondo').
candidato(cs, 'Edmundo Bal').
candidato(pp, 'Isabel Díaz Ayuso').
candidato(vox, 'Rocío Monasterio').

izquierda(up).
izquierda(mm).
izquierda(psoe).

derecha(cs).
derecha(pp).
derecha(vox).

% 1 sería más a la izquierda y 6 más a la derecha
espectro(up, 1).
espectro(mm, 2).
espectro(psoe, 3).
espectro(cs, 4).
espectro(pp, 5).
espectro(vox, 6).

% TODO: En caso de que algún partido minoritario saque algún escaño el 4 de

```

```
%      mayo se añadirá.
```

```
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% Los partidos se encuentran en el mismo espectro o no                                %
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
```

```
mismo_espectro(Partidos) :-
    todos_derecha(Partidos),!;
    todos_izquierda(Partidos).
```

```
distinto_espectro(Partidos) :- not(mismo_espectro(Partidos)).
```

```
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% Solicitar resultado de un partido                                                    %
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
```

```
pedir_escanos(Partido) :-
    write('Introduce el número de escaños que ha obtenido '),
    nombre(Partido, S),
    write(S),
    nl,
    read(X),
    assert(escanos(Partido, X)).
```

```
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% Reiniciar resultados                                                                %
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
```

```
reiniciar_escanos :-
    forall(escanos(X,Y),retract(escanos(X,Y))),
    retract(lista_partidos(_)).
```

```
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% Dice si un partido tiene o no escaños                                              %
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
```

```
sin_escanos(Partido) :-
    escanos(Partido,0).
```

```
con_escanos(Partido) :-
    escanos(Partido,X),
```

X >= 1.

```

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% Lista los partidos con escanos
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
anadir_partido(P) :-
    lista_partidos(X),
    retract(lista_partidos(X)),
    append(X,[P],L),
    assert(lista_partidos(L)).

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% Ordenado rápido de partidos
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% NOTE: Esto debería estar en util.pl pero por algún motivo no quiere funcionar
%      ahí
quicksort([P|Ps], R) :-
    partition(Ps,P,Izq,Der),
    quicksort(Izq,IR),
    quicksort(Der,DR),
    append(IR,[P|DR],R).

quicksort([], []).

partition([P|Ps],Pivot,[P|IR],DR) :-
    escanos(P,PX),
    escanos(Pivot,PivotX),
    PX > PivotX,!,
    partition(Ps,Pivot,IR,DR).

partition([P|Ps],Pivot,IR,[P|DR]) :-
    escanos(P,PX),
    escanos(Pivot,PivotX),
    (PX < PivotX; PX == PivotX),!,
    partition(Ps,Pivot,IR,DR).

partition([],_,[], []).

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% Calculo de escaños totales de un grupo de partidos
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

```

```

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
escanos_totales([Partido|[]], S) :-
    escanos(Partido,X),
    S is X,! .

escanos_totales([Partido | Partidos], S) :-
    escanos(Partido, X),
    escanos_totales(Partidos, Y),
    S is X + Y.

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% Comprobación de que todos los partidos son del mismo espectro político %
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
todos_derecha(Partidos) :-
    forall(member(X,Partidos),derecha(X)).

todos_izquierda(Partidos) :-
    forall(member(X,Partidos),izquierda(X)).

```

pactos.pl

```

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
%
%   pactos.pl   -   El pactometro 3.0 (Tiembra Ferreras).
%
%
% author: Martín "n3m1.sys" Romera Sobrado
%
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

:- [partidos].

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% Reglas sobre pactos. Estas reglas se basan en las intenciones de pactos de %
% los partidos según lo que han dicho en los medios
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
pactaria_con(up,up).
pactaria_con(up,mm).
pactaria_con(up,psoe).
pactaria_con(mm,mm).
pactaria_con(mm,up).

```



```

pactaria_con(mm,psoe).
pactaria_con(psoe,psoe).
pactaria_con(psoe,mm).
pactaria_con(psoe,cs).
pactaria_con(cs,cs).
pactaria_con(cs,psoe).
pactaria_con(cs,pp).
pactaria_con(cs,vox).
pactaria_con(pp,pp).
pactaria_con(pp,cs).
pactaria_con(pp,vox).
pactaria_con(vox,vox).
pactaria_con(vox,cs).
pactaria_con(vox,pp).

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% Definición de mayoría parlamentaria para hacer posible un pacto %
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% Los partidos de X forman una mayoría (la mitad más 1) de los escaños, es decir
% si los partidos de X conforman al menos 70 escaños
mayoria(Partidos) :-
    escanos_totales(Partidos, S),
    S >= 69.

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% Valoriación de un pacto %
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
valorar_pacto(Ps) :-
    eliminar_repes(Ps, Parts),
    eliminar_sin_escanos(Parts, Partidos),
    (
        mayoria(Partidos),
        (
            (
                % Solo hay un partido en el pacto
                solo_uno(Partidos),!,
                comen_mayoria_absoluta(Partidos)
            );
            (
                % Todos los partidos quieren pactar entre si

```

```

        not(solo_uno(Partidos)),
        pactos_bidireccionales(Partidos),!,
        comen_pactos_bidireccionales(Partidos)
    );(
        % Algunos partidos no quieren pactar con otros que a su vez
        % quieren pactar con ellos
        not(solo_uno(Partidos)),
        not(pactos_bidireccionales(Partidos)),
        pactos_unidireccionales(Partidos),!,
        comen_pactos_unidireccionales(Partidos)
    );(
        % Hay partidos que no quieren votar entre ellos
        not(solo_uno(Partidos)),
        not(pactos_bidireccionales(Partidos)),
        not(pactos_unidireccionales(Partidos)),!,
        comen_no_pactos(Partidos)
    )
);
% No cumplen la condición de mayoría
not(mayoria(Partidos)),
comen_no_mayoria(Partidos)
).

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% Evaluación de cantidad de partidos en una agrupación %
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
solo_uno([_|[]]).

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% Evaluación de intención de pacto %
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
pactos_bidireccionales(Partidos) :-
    forall(member(X,Partidos),pacta_resto(X,Partidos)).

pacta_resto(X,Partidos) :-
    forall(member(Y,Partidos),pactaria_con(X,Y)).

pactos_unidireccionales(Partidos) :-
    forall(member(X,Partidos),pacta_uniresto(X,Partidos)).

```

```

pacta_uniresto(X,Partidos) :-
    forall(member(Y,Partidos),pacta_alguien(X,Y)).

pacta_alguien(X,Y) :-
    pactaria_con(X,Y);
    pactaria_con(Y,X).

```

util.pl

```

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
%                                                                    %
%   util.pl   -   Utilidades                                                                    %
%                                                                    %
%   author:   Martín "n3m1.sys" Romera Sobrado                                                                    %
%                                                                    %
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% Cabeza de la lista                                                                    %
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
cabeza([L|_],L).

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% Calcula la máxima distancia entre los partidos de una lista de partidos %
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
max_distancia(Partidos, X) :-
    sort_izqder(Partidos, IzqDer),
    sort_derizq(Partidos, DerIzq),
    cabeza(IzqDer, MasIzq),
    cabeza(DerIzq, MasDer),
    espectro(MasIzq, EspIzq),
    espectro(MasDer, EspDer),
    X is EspDer - EspIzq.

sort_izqder(L,S) :-
    permutation(L,S),
    izqder(S),!.

sort_derizq(L,S) :-

```

```

    permutation(L,S),
    derizq(S),!.

izqder([]).
izqder([_]).
izqder([X,Y|T]) :-
    espectro(X,EX),
    espectro(Y,EY),
    EX < EY,
    izqder([Y|T]).

derizq([]).
derizq([_]).
derizq([X,Y|T]) :-
    espectro(X,EX),
    espectro(Y,EY),
    EX > EY,
    derizq([Y|T]).

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% Elimina los partidos que no han obtenido escanos de un grupo de partidos %
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
eliminar_sin_escanos(Partidos, Resultado) :-
    assert(partidos(Partidos)),
    forall(sin_escanos(X), eliminar(X)),
    partidos(Resultado),
    retract(partidos(Resultado)).

eliminar(X) :-
    partidos(Partidos),
    retract(partidos(Partidos)),
    eliminarTodos(X, Partidos, R),
    assert(partidos(R)).

eliminarTodos(_, [], []).
eliminarTodos(X, [X|Xs], Y) :-
    eliminarTodos(X, Xs, Y).
eliminarTodos(X, [T|Xs], [T|Y]) :-
    dif(X, T),
    eliminarTodos(X, Xs, Y).

```

```

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% Elimina los partidos repetidos en un grupo de partidos %
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
eliminar_repes([], []).

```

```

eliminar_repes([H | T], List) :-
    member(H, T),
    eliminar_repes(T, List),!.

```

```

eliminar_repes([H | T], [H|T1]) :-
    \+member(H, T),
    eliminar_repes(T, T1),!.

```

comentarios.pl

```

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
%
% comentarios.pl - Contiene los comentarios del periodista %
%
% author: Martín "n3m1.sys" Romera Sobrado %
%
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

```

```

:- [partidos].

```

```

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% Listar partidos %
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

```

```

listar_partidos([]).

```

```

listar_partidos([P]) :-
    nombre(P,N),
    write('y '),
    write(N).

```

```

listar_partidos([P|Ps]) :-
    nombre(P,N),
    write(N),
    write(', '),
    listar_partidos(Ps).

```

```

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% Comentario de quien ha ganado las elecciones %
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
ganador(Partido) :-
    escanos(Partido, X),
    PRG is X mod 3,
    nombredet(Partido, Nombre),
    candidato(Partido, Candidato),
    % Selección de frase pseudoaleatoria según el número de escaños
    (
        (PRG == 0,
            write('Con estos resultados, '),
            write(Candidato),
            write(' habría ganado las elecciones.'),
            !
        );
        (PRG == 1,
            write('Ahora mismo con los resultados que tenemos, '),
            write(Nombre),
            write(' tendría mayoría parlamentaria en la Asamblea de Madrid.'),
            !
        );
        (PRG == 2,
            write(Nombre),
            write(' con '),
            write(X),
            write(' escaños, se sitúa como primera fuerza parlamentaria.'),
            !
        )
    ).

```

```

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% Comentario si un conjunto de partidos no cumple la condición de mayoría %
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
comen_no_mayoria(Partidos) :-
    escanos_totales(Partidos, Escanos),
    PRG is Escanos mod 4,
    (
        (PRG == 0,

```

```

        write('Si querías un pacto entre esos partidos... parece que se'),
        write('quedan cortos. Prueba dentro de 4 años :)'),nl,!
    );
    (PRG == 1,
        write('No se si lo sabías, pero '),
        write(Escanos),
        write(' es menos que 69 :/.'),nl,!
    );
    (PRG == 2,
        write('No voy ni a intentar pensar si se van a llevar bien... ¡Ni!'),
        write('siquiera forman mayoría!'),nl,!
    );
    (PRG == 3,
        write('Antes de mandarme hacer algo inútil ¿Qué tal si haces la '),
        write('suma antes?'),nl,!
    )
).

```

```

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% Comentario en caso de mayoría absoluta %
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

```

```

comen_mayoria_absoluta(Partido) :-
    escanos(Partido,Escanos),
    candidato(Partido,Presidente),
    write('Contra todo pronóstico, '),
    write(Presidente),
    write(' lograría la mayoría absoluta con los resultados actuales con '),
    write(Escanos),
    write(' escaños. '),nl.

```

```

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% Comentario si los partidos tienen opción a pacto bidireccional %
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

```

```

comen_pactos_bidireccionales(Partidos) :-
    escanos_totales(Partidos, Escanos),
    (
        (mismo_espectro(Partidos),
            % Los partidos son del mismo espectro politica
            % (o derecha o izquierda)
            (todos_derecha(Partidos),

```

```
    % Los partidos son todos de derechas
    write('Una coalición entre '),
    listar_partidos(Partidos),
    write(' daría paso a un gobierno de derechas en la Asamblea'),
    write(' de Madrid liderado por '),
    quicksort(Partidos,Ordenados),
    cabeza(Ordenados,Lider),
    candidato(Lider,Presidente),
    write(Presidente),
    write(' con un respaldo de '),
    write(Escanos),
    write(' escaños. '),nl
);
(todos_derecha(Partidos),
    % Los partidos son todos de izquierdas
    write('Una coalición que podría dar paso a un gobierno del'),
    write(' cambio liderado por'),
    quicksort(Partidos,Ordenados),
    cabeza(Ordenados,Lider),
    candidato(Lider,Presidente),
    write(Presidente),
    write(' con el respaldo de los '),
    write(Escanos),
    write(' escaños de '),
    listar_partidos(Partidos),
    write(' '),nl
)
);
(distinto_espectro(Partidos),
    % Los partidos son de derechas e izquierdas
    listar_partidos(Partidos),
    write(' suman juntos una mayoría mixta de centro derecha y centro'),
    write(' izquierda con '),
    write(Escanos),
    write(' escaños que habilitarían un gobierno liderado por '),
    quicksort(Partidos,Ordenados),
    cabeza(Ordenados,Lider),
    candidato(Lider,Presidente),
    write(Presidente),
    write(' '),nl
```



```

    )
).

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% Comentario si los partidos pueden optar a pactos unidireccionales %
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
comen_pactos_unidireccionales(Partidos) :-
    % Por la forma en la que estan definidas las intenciones de pacto, esto solo
    % puede ocurrir dentro de la izquierda de forma que para simplificar solo se
    % definirán los dialogos de esta situación para la izquierda
    escanos_totales(Partidos, Escanos),
    (
        (todos_izquierda(Partidos),
            write('Un pacto entre '),
            listar_partidos(Partidos),
            write(' podrían dar paso a un gobierno progresista liderado por '),
            quicksort(Partidos,Ordenados),
            cabeza(Ordenados,Lider),
            candidato(Lider,Presidente),
            write(Presidente),
            write(' con '),
            write(Escanos),
            write(' escaños de respaldo. Sin embargo '),
            candidato(psoe, CPSOE),
            write(CPSOE),
            write(' ha advertido que no tenía intención de pactar con '),
            candidato(up, CUP),
            write(CUP),
            write(' lo que podría bloquear esta posibilidad. '),nl
        );
        (not(todos_izquierda(Partidos)),
            write('Hmmm... Algo falla porque esto no debería de poder ocurrir'),
            write(' según la lógica del vago de mi programador. '),nl
        )
    ).

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% Comentario si los partidos no quieren pactar entre ellos %
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
comen_no_pactos(Partidos) :-

```

```
% Esta situación solo se puede dar entre partidos de diferentes espectros
% políticos, así que el mensaje se decidirá mediante la diferencia máxima
% (la distancia en el espectro entre los dos extremos) de los partidos
max_distancia(Partidos,Dist),
```

```
(
  (Dist == 2, !,
    write('Un poco extraña la combinación de '),
    listar_partidos(Partidos),
    write(', ¿no crees?. Es decir sí, suman mayoría y eso pero...'),
    write('¿No hay nada mejor?'),nl
  );
  (Dist == 3, !,
    write('No me imagino un mundo en el que '),
    listar_partidos(Partidos),
    write('pacten... Prueba otra cosa anda...'),nl
  );
  (Dist == 4, !,
    write('¿Sabes eso de que no hay que beber antes de conducir? '),
    write('Pues aplicate lo mismo para cuando vayas a preguntarme'),
    write(' por la posibilidad de un pacto. '),nl
  );
  (Dist == 5, !,
    write('Tú fumas crack'),nl
  )
).
```

```
% %%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% % Comentario por si se encuentra un partido que no tiene escaños en la %
% % combinación propuesta %
% %%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% comen_sin_escanos(P) :-
%   write('No creo que nadie vaya a pactar con un partido sin escaños como '),
%   nombre(P,N),
%   write(N),
%   write(' '),nl.
```

```
% %%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% % Comentario por si se encuentra un partido repetido en un grupo de partidos %
% %%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% comen_repe(P) :-
```

```
%   write('Por mucho que te guste '),  
%   nombre(P,N),  
%   write(N),  
%   write(', no va a tener más escaños de los que tiene. '),nl.
```